
Chapter 7

Advanced Type Systems

Type structure is a syntactic discipline for enforcing levels of ab-
straction. . . .What computation has done is to create the necessity
of formalizing type disciplines, to the point where they can be en-
forced mechanically.

— John C. Reynolds
Types, Abstraction, and Parametric Polymorphism [?]

This chapter examines some advanced type systems and their uses in functional
and meta-languages. Specifically, we consider the important concepts of parametric
polymorphism, subtyping, and intersection types. Their interaction also leads to new
problems which are subject of much current research and beyond the scope of these
notes.

Each language design effort is a balancing act, attempting to integrate mul-
tiple concerns into a coherent, elegant, and usable language appropriate for the
intended application domain. The language of types and its interaction with the
syntax on the one hand (through type checking or type reconstruction) and the
operational semantics on the other hand (through type preservation and a progress
theorem) is central. Types can be viewed in various ways; perhaps the most impor-
tant dichotomy arises from the question whether types are an inherent part of the
semantics, or if they merely describe some aspects of a program whose semantics is
given independently. In one view types are intrinsic to the terms in a language, in
another they are extrinisic.

This is related to the issue whether a language is statically or dynamically typed.
In a statically typed language, type constraints are checked at compile-time, before
programs are executed; only well-typed programs are then evaluated. It also carries
the connotation that types are not maintained at run-time, the type-checking having
guaranteed that they are no longer necessary. In a dynamically typed language,

208 CHAPTER 7. ADVANCED TYPE SYSTEMS

types are instead tested at run-time in order to prevent meaningless operations
(such as trying to compute fst z).

The semantics of Mini-ML we have presented it so far is essentially untyped, that
is, the operational semantics is given for untyped expressions. Since the language
satisfies type preservation, a type expresses a property of untyped programs. A
program may have multiple types, since it may satisfy multiple properties express-
ible in the type system. Typing was intended to be static: programs are checked
for validity and then evaluated. On the other hand, the semantics of LF (and,
by extension, Elf) is intrinsic: only well-typed objects represent terms or deriva-
tions. Furthermore, types occur explicitly in objects and are present at run-time,
since computation proceeds by searching for an object of a given type. Nonetheless,
typing is static in the sense that only well-typed queries can be executed.

Which criteria should be used to evaluate the relative merits of type systems?
Language design is an art as well as a craft, so there are no clear and unambiguous
rules. Nevertheless, a number questions occur frequently in the comparison of type
systems.

Generality. How many meaningful programs are admitted? For example, the
typing discipline of Mini-ML prohibits self-application (λx. x x), even though this
function could be meaningfully applied to many arguments, including, for example,
the identity function.

Accuracy. To what level of detail can we describe or prescribe the properties of
programs? For example, in Mini-ML we can express that f represents a partial
function from natural numbers to natural numbers. We can not express that it
always terminates or that it maps even numbers to odd numbers.

Decidability. Is it decidable if a given expression is well-typed? In a statically
typed language this is important, since the compiler should be able to determine if
a given input program is meaningful and either execute or reject it.

Often, generality, accuracy, and decidability conflict: if a system is too general or
too accurate it may become undecidable. Thus these design goals must be traded
off against each other. Other criteria help in making such choices.

Brevity. How much type information must be provided by the programmer? It is
desirable for the programmer to be allowed to omit some types if there is a canonical
way of reconstructing the missing information. Conversely, decidability can often
be recovered for very general type systems by requiring many type annotations. In
the case of Mini-ML, no type information is necessary in the input language; in Elf,
a program consists almost entirely of types.

Elegance. Complex, convoluted type systems may be difficult to use in practice,
since it may be hard for the programmer to predict which expressions will be con-
sidered valid. It also becomes more difficult for the type-checker to give adequate
feedback about the source of a type error. Furthermore, a type system engenders
a certain discipline or style of programming. A type system that is hard to under-
stand endangers the elegance and simplicity and therefore the usability of the whole
programming language.

Efficiency. Besides the theoretical property of decidability, the practical ques-
tion of the efficiency of type checking must also be a concern. The Mini-ML
type-checking problem is decidable, yet theoretically hard because of the rule for
let name. In practice, type checking for ML (which is a significant extension of
Mini-ML along many dimensions) is efficient and takes less than 10% of total com-
pilation time except on contrived examples. The situation for Elf is similar.

We return to each of these critera in the following sections.

7.1 Parametric Polymorphism

Parametric polymorphism, like many other type constructs, can be viewed from a
number of perspective. In keeping with the development in Chapter ?? we start
with the extrinsic view: types are assigned to typeless expressions. We generalize
the language of types (but not the language of expressions) to include types of the
form ∀α. τ . An expression should have this type if it has all instance types [τ ′/α]τ .
As usual, we express this parametrically in the meta-language: e has type ∀α. τ if
e has type τ where α is a parameter.

Types τ ::= . . . | ∀α. τ

Now the rule which introduces the new form of type in the conclusion. As in
natural deduction (see Chapter ??), we therefore call this an introduction rule.

∆ . e : τ
tp alliα

∆ . e : ∀α. τ

We also refer to this rule as universal generalization. The deduction of the
premise must be parametric in α. In particular, α may not already occur in ∆,
which would lead to a system that does not satisfy type preservation or progress
(see Exercise ??).

Since the premise is indeed parametric in α, we can substitute an arbitrary type
τ ′ for α to obtain another valid typing for e from the same assumptions ∆.

∆ . e : ∀α. τ
tp alle

∆ . e : [τ ′/α]τ

Since this rule eliminates the universal quantifier from the judgment in the
premise, it is called an elimination rule. We also refer to it as universal instantiation.

It is easy to extend our LF representation of the language and the typing rules to
encompass parametric polymorphism as defined above. Note that the language of
expressions and evaluation only change in that we eliminate let name. Once again
we use the idea of higher-order abstract syntax, this time at the level of Mini-ML
types, in order to represent the variable binder ∀α.

all : (tp -> tp) -> tp.

tp_alli : of E (all [a] T a)

<- ({a:tp} of E (T a)).

tp_alle : {T’:tp} of E (T T’)

<- of E (all [a] T a).

Note once again how application T T’ is used to represent substitution [τ ′/α]τ .
It is also worth noting the explicit quantifier on T’ in the last clause. In LF, this
clause would read

tp alle : ΠE:exp. ΠT :tp→ tp.
ΠT ′:tp. of E (all (λa:tp. T a))→ of E (T T ′).

The reason to make the quantifier on T ′ explicit in Elf (rather than leaving it
implicit as the quantifiers on E and T ) is because there is no occurrence of T ′ as
an argument to a constant. As a result, T ′ would often remain ambiguous when
the constant tp_alle is used. In technical terms, T ′ has no strict occurrence in the
body of the clause, which generally means that T ′ should be explicitly quantified.

Unfortunately, the typing rules for Mini-ML with parametric polymorphism in
extrinsic form are no longer syntax-directed. The rules tp alli and tp alle are always
potentially applicable, since the expression e does not change. As a result, the
LF signature can no longer be used for type inference. It is quite likely that type
inference for this language is no longer decidable. Although we currently do not
have a proof of this fact, two closely related systems mentioned later have been
shown to be undecidable.

As a simple example, consider the identity function lam x. x which can be seen



7.1. PARAMETRIC POLYMORPHISM 211

to have type ∀α. α→ α.

tp var
x:α . x : α

tp lam
· . lam x. x : α→ α

tp alliα

· . lam x. x : ∀α. α→ α

This derivation is represented in Elf by

_ : of (lam [x] x) (all [a] a => a)

= tp_alli [a:tp] tp_lam [x:exp] [u:of x a] u.

Note that we can no longer rely on the operational semantics to construct the typing
derivation, so we give it explicitly as an anonymous definition.

Another example is self-application, lam x. x x, a function that could not be
typed in the formulation of Mini-ML in Chapter ??. However, with the use of
parametric polymorphism we can give it type

(∀α. α→ α)→ (∀α. α→ α)

with the following derivation, where we use the abbreviation one = ∀α. α→ α:

tp var
x:one . x : one

tp alle
x:one . x : one → one

tp var
x:one . x : one

tp app
x:one . x x : one

tp lam
· . lam x. x x : one → one

Again, we can check this derivation in Twelf.

one : tp = all [a] a => a.

_ : of (lam [x] app x x) (one => one)

= tp_lam [x] [u:of x one] tp_app u (tp_alle one u).

However, this is not the only type for self-application (see Exercise ??).
The proof of type preservation needs to be modified since inversion on the typing

derivation no longer yields a unique rule (the expression e does not change in the
rules for polymorphic generalization and instantiation). We account for this by using
a nested (or lexicographic) induction on the evaluation and the typing derivation.

Theorem 7.1 (Type Preservation with Extrinsic Parametric Polymorphism)
If e ↪→ v and · . e : τ then · . v : τ .

Proof: By nested induction on the structure of D :: e ↪→ v and P :: · . e : τ . That
is, we can appeal to the induction hypothesis for a smaller evaluation and arbitrary
typing derivation, or an identical evaluation and smaller typing derivation.

Case: D is arbitrary and

P =

P1

· . e : τ1
tp alliα.

· . e : ∀α. τ1
Then

· . v : τ ′ By i.h. on D and P1

· . v : ∀α. τ1 By rule tp alliα

Case: D is arbitrary and

P =

P1

· . e : ∀α. τ1
tp alle.

· . e : [τ ′/α]τ1
Then

· . v : ∀α. τ1 By i.h. on D and P1

· . v : [τ ′/α]τ1 By rule tp alle

Cases: In all other cases for D we can now exclude the possibility that P ends in
tp alli or tp alle which is already handled above. Therefore we literally copy
all cases in the original proof of type preservation for Mini-ML (Theorem 2.5).

2

Before we move on to other formulations of parametric polymorphism, we con-
sider the relationship between the original formulation of Mini-ML in which poly-
morphism was captured by a letname-construct, and the second one above in
which polymorphism is explicit in the type. It is clear from the example of self-
application, lamx. x x that the system above permits some terms that are not
typable using let name.

In order to show that explicit polymorphism is strictly more general, we would
have to show that every use of let name can be eliminated. One way to do this
without changing the operational behavior is to simply reduce let namex = e1 in e2

to [e1/x]e2. However, this explodes program size and is not modular. Can we replace
every occurrence of let name by an appropriate form of letval, taking advantage
of explicit polymorphism? The answer if affirmative. The key is again is the notion
of parametric judgment. For example, if ∆ . e : α → α where α does not occur
in ∆, then ∆ . e : τ → τ for all types τ . It turns out that Mini-ML without
explicit polymorphism has the property that each expression e that is well-typed
has a principal type τ with the following properties:

1. ∆ . e : τ , that is, τ is a type of e. Let α1, . . . , αn be the free type variables of
τ that do not occur in ∆.

2. ∆ . e : [σ1/α1, . , σn/αn]τ for any types σ1, . . . , σn.

3. If ∆ . e : σ then σ = [σ1/α1, . , σn/αn]τ for some σ1, . . . , σn.

Part 2 follows from part 1 directly since the judgment ∆ . e : τ is parametric
in α1, . . . , αn. Part 3 is a deeper property which requires a relatively complex
development for its proof which is beyond the scope of this book. Briefly, the
lines of reasoning are as follows. The rules are syntax-directed, but they do not
determine the types uniquely. Instead they impose some equational constraints on
the types. For example, in an expression e1 e2, the argument type of e1 must match
the type of the argument, e2. Such equations always have a most-general solution:
every solution to equational constraints can be obtained as an instance of this most
general solution. Technically, this problem is called unification, the property above
is the existence of most general unifiers. For more on this problem of typing, see [?].

7.2 Intrinsic Polymorphism

Parametric polymorphism, like many other type constructs, can be viewed from
a number of perspectives. In the preceding section we viewed types as extrinsic
to expressions; here we view them as intrinsically part of the terms. Rather than
a pure type assignment ∆ . e : τ where a given expression e may have many
different types, our main judgment ∆ . M : τ has the property that the term M
determines its type uniquely. This requires that terms M mention some types to
remove ambiguity, for example, in the identity function lam x. x.

For parametric polymorphism this means that polymorphic functions are ex-
plicitly parameterized over a type. We write such an abstraction as Λα, as in
Λα. lamx:α. x for the polymorphic identity function. It takes two arguments: first
a type τ then an argument v of type τ . For the sake of symmetry we must then
also have a means to explicitly apply a polymorphic function to a type. Write this
this as M [τ ].

We then have the rules

∆ . M : τ
tp alliα

∆ . Λα. M : ∀α. τ

∆ . M : ∀α. τ
tp alle

∆ . M [τ ′] : [τ ′/α]τ

The rest of the language can be extended to enforce uniqueness of types via a mode

analysis of the inference rules for type assignment. For example, in the rule

∆, x:τ1 . e : τ2
tp lam

∆ . lam x. e : τ1 → τ2

the type τ1 that is added to the context ∆ is not uniquely determined by e. There-
fore, we need to decorate the lam-abstraction with τ1.

∆, x:τ1 . M : τ2
tp lam

∆ . lam x:τ1M : τ1 → τ2

These considerations yield the following language of terms.

Terms M ::= z | s M | (case M1 of z⇒M2 | s x⇒M3) Natural numbers
| 〈M1,M2〉 | fst M | snd M Pairs
| lam x:τ. M |M1 M2 Functions
| letval x = M1 in M2 Definitions
| fix x:τ. M Recursion
| x Variables
| Λα. M |M [τ ] Polymorphism

Note that the introduction of let name is not only redundant (in the sense that it
solves the same problem as explicit polymorphism), but that it no longer works cor-
rectly: the typing rule for let name relied on the property that the same expression
can have more than one type. We omit the full set of typing rules, which are com-
pletely straightforward modifications of the type assignment rules for expressions.

The two examples above now have the following form:

Λα. lamx:α. x : ∀α. α→ α
lamx:∀α. α→ α. x [∀α. α→ α] x : (∀α. α→ α)→ (∀α. α→ α)

Before we consider the operational semantics, we prove the uniqueness of types.
As always, we include renaming of bound variables when considering equality. Here
the variables are type variables α and ∀α is the only binding construct.

Theorem 7.2 (Uniqueness of Intrinsic Polymorphic Types)
If ∆ . M : τ and ∆ . M : σ then τ = σ.

Proof: By simultaneous induction over the structure of P :: ∆ . M : τ and Q ::
∆ . M : σ. Note that we already assume that the context ∆ is the same on both
sides. We show only one case.

Case:

P =

P2

∆, x:τ1 . M2 : τ2
tp lam

∆ . lam x:τ1. M2 : τ1 → τ2

and Q =

Q2

∆, x:σ1 . M2 : σ2
tp lam

∆ . lam x:σ2. M2 : σ1 → σ2

Then

lam x:τ1. M2 = M = lam x:σ1. M2 P and Q end in M
τ1 = σ1 by equality
(∆, x:τ1) = (∆, x:σ1) by equality
τ2 = σ2 by i.h. on P2 and Q2

τ1 → τ2 = σ1 → σ2 by equality

2

The implementation of this proof in Twelf is somewhat awkward, since we cannot
access the internal (judmental) notion of equality in the framework. Therefore
we define a structural equality on types. We show below the implementation of
the fragment containing only function types. The reader may consult the on-line
material for the remaining cases. First, type equality.

eq : tp -> tp -> type. %name eq E e.

eq_arrow : eq (arrow T1 T2) (arrow S1 S2)

<- eq T1 S1

<- eq T2 S2.

We need a lemma stating that type equality is reflexive, which follows by in-
duction on the structure of the type. Note that there is a dependency between the
input and output argument of refl.

refl : {T:tp} eq T T -> type.

%mode refl +T -E.

refl_arrow : refl (arrow T1 T2) (eq_arrow E2 E1)

<- refl T1 E1

<- refl T2 E2.

%terminates T (refl T _).

Next we prove the uniqueness of the types, by simultaneous induction on P and
Q.

% Uniqueness of types

tpu : of* E T -> of* E S -> eq T S -> type.

%mode tpu +P +Q -E.

tpu_lam : tpu (tp_lam* P2) (tp_lam* Q2) (eq_arrow E2 E1)

<- refl T1 E1

<- ({x:term*} {u:of* x T1}

tpu u u E1 -> tpu (P2 x u) (Q2 x u) E2).

tpu_app : tpu (tp_app* P2 P1) (tp_app* Q2 Q1) E1

<- tpu P1 Q1 (eq_arrow E1 E2).

%terminates [P Q] (tpu P Q _).

The operational semantics extends quite easily from the type assignment system;
only the case of type abstraction and application require some consideration. It may
be possible, for example, to evaluate underneath a type abstraction with the rule

M ↪→ V

Λα. M ↪→ Λα. V

where the evaluation of the premise would be parametric in α. However, just as we
only evaluate closed expressions, it seems most consistent to evaluate only closed
terms. This is because only the meaning of closed terms is given unambiguously.
Since types are intrinsic to terms, this would naturally preclude both free type and
free term variables. We therefore instead use the rule

ev tlam
Λα. M ↪→ Λα. M

Type application is straightforward.

M ↪→ Λα. M ′ [τ/α]M ′ ↪→ V
ev tapp

M [τ ] ↪→ V

Here we use V for term values which are slightly different from expression values v
in that they include a case for type abstractions.

Term Values V ::= z | s V Natural numbers
| 〈V1, V2〉 Pairs
| lam x. M Functions
| Λα. M Polymorphism

We relate terms to expression via erasure and write |M | = e. The definition is
obvious, except for the cases of type abstraction and application, where we define

|Λα. M | = |M |
|M [τ ]| = |M |

Note that erasure forgets structure: many different terms erase to the same ex-
pression. However, there is a bijection between typing derivations of terms and
expressions. This is implicit in the proof of the following theorem which asserts the
correctness of erasure.

Theorem 7.3 (Relation between Extrinsic and Intrinsic Typing)

1. If ∆ . e : τ then there is an M such that |M | = e and ∆ . M : τ .

2. If ∆ . M : τ and |M | = e then ∆ . e : τ .

Proof: In each direction by a proof over the structure of the given derivation. 2

Both erasure and its correctness proof are easy to implement and elided here.
They can be found in1.

Unfortunately, the translation developed above does not preserve the operational
semantics. A simple counterexample is the term

. Λα. fix x:α. α : ∀aα. α

It evaluates to itself in one step while its erasure, fix x. x does not have a value. At
some cost in elegance, this can be repaired by inserting vacuous abstractions and
applications into the translation. Assuming we have the unit type 1 (see ??) and
the unit element 〈 〉, we change erasure to

|Λα. M | = λx. |M | x not free in M
|M [τ ]| = |M | 〈 〉

However, the translation is no longer a bijection because the vacuous variable x can
be assigned an arbitrary type. Furthermore, an expression such as fix x. x at type
∀α. α is not in the image of the translation function.

Another solution is to restrict universal generalization on expression or type
abstraction on terms to values. This is an instance of the so-called value restriction
which has been adopted for Standard ML [?] for related reasons. The modified rules
are

∆ . v : τ
tp alliα

∆ . v : ∀α. τ
∆ . V : τ

tp tlamα

∆ . Λα. V : ∀α. τ

Under these restricted rules we have a strong correspondence theorem illustrated
by the following picture.

1[poly/intrinsic.elf ]

e v

M V

-

-

6 6

e ↪→ v

M ↪→ V

e = |M | v = |V |

This decomposes into two theorems. The first is an instance of coherence. We
think of e only as a shorthand for an intrinsically typed term M . In that case the
meaning of e is ambiguous, since different typing derivations for e yield different
terms M which evaluate to different answers V . Coherence states that it does not
matter how we reconstruct the missing types in e, because whatever answers V we
get by this path erase to the same v. This independence of the answer from a typing
derivation is the meaning of coherence; this is particularly simple instance of the
idea.

The second theorem shows that an untyped operational semantics is adequate:
we do not need to carry types at run-time. If we erase M to obtain e and then
evaluate e we obtain the same value as if we evaluated M according to the typed
semantics and then erased the types from the answer. Since the types in the answer
are not obvservable (in values they can appear only under lam-abstractions which
are not observable), the erased value v is observably equivalent to V .

The proof of these theorems is left as Exercise ?? Note that the value restriction
is critical for this correspondence.

7.3 Indexed Representation of Intrinsic Typing

Intrinsic types are considered to be an integral part of terms and a term that is not
well typed is considered meaningless. For example, in an intrinsically typed system
we would not ask the question if fst 〈z, (λx. sndz)〉 has a value, because it is not
well typed. In an extrinsically typed system we may note that it is not well-typed,
but we can nonetheless state that it evaluates to z.

In that sense our development above falls somewhat short of a fully intrinsically
typed language: term M contain types, but the evaluation judgment still could be

applied to terms that do not type check. The type of a term is unique if it exists,
but it need not exist.

The logical framework allows us to explore an even stronger version of intrinsic
typing were the types are directly part of the representation. Because of its hidden
complexity, such formulations are not particularly popular in informal presentation.
We would annotate a term with its (unique) type and the presentation of the syntax
itself carries this type. When a term M intrinsically has type τ , we write M τ . The
syntax for such terms (showing only the functional fragment) would be:

Terms Mτ ::= znat | (s Mnat)nat | (case Mnat
1 of z⇒Mτ

2 | s xnat ⇒Mτ
3 )τ

| 〈Mτ
1 ,M

σ
2 〉τ×σ | (fst Mτ×σ)τ | (snd Mτ×σ)σ

| (lam xτ . Mσ)τ→σ | (Mσ→τ
1 Mσ

2 )τ

| (let val xσ = Mσ
1 in Mτ

2 )τ

| (fix xτ . Mτ )τ

| xτ
| (Λα. Mτ )∀α. τ | (M∀α. σ [τ ])[τ/α]σ

The implementation of this formulation in LF is particularly elegant, since we
can index terms by there type. The adequacy theorem for the representation than
asserts a compositional bijection between well-typed terms in Mini-ML and well-
typed LF objects of the representation type.

term : tp -> type. %name term M x.

z’ : term nat.

s’ : term nat -> term nat.

case’ : term nat -> term T -> (term nat -> term T) -> term T.

pair’ : term T -> term S -> term (T * S).

fst’ : term (T * S) -> term T.

snd’ : term (T * S) -> term S.

lam’ : (term T -> term S) -> term (T => S).

app’ : term (S => T) -> term S -> term T.

letv’ : term S -> (term S -> term T) -> term T.

fix’ : (term T -> term T) -> term T.

tlam’ : ({a:tp} term (T a)) -> term (all [a] T a).

tapp’ : term (all [a] S a) -> {T:tp} term (S T).

This form of indexed representation of terms has a number of advantages when
it is possible. For one, no explicit type-checking rules need to be implemented: the

framework implementation itself will accomplish this. In practice, however, since
the type indices remain implicit, this means the framework will have to perform a
complicated form of type reconstruction. Sometimes it will not be able to decide
whether the term can be typed. For example, the anonymous definition

one = all [a] a => a.

_ = lam’ [x] app’ (tapp’ x one) x.

yields the error

Typing ambiguous -- unresolved constraints

T1 one = all ([a:tp] T1 a) => T2.

This means we have to supply more type information. The analysis of intrinsic
typing in Section 7.2 shows what is necessary for types to be unique. This translates
to a corresponding criterion on the Elf representation. Here it means that if we
provide the type of the variable x, all other types are determined.

one = all [a] a => a.

_ = lam’ [x:term one] app’ (tapp’ x one) x.

Another advantage is that some theorems, such as Mini-ML type preservation,
are now also intrinsic to the representation and do not need to be proved separately.
We declare

eval’ : term T -> term T -> type.

%mode eval’ +M -V.

so that type-checking the declarations for eval’ will automatically verify type
preservation. Informally, the corresponding judgment could be written as M τ ↪→
V τ , capturing the same constraint. Since T is an implicit argument, the evaluation
rules are textually the same as for expressions in Mini-ML, although the result of
reconstruction is different. As an example we show the rule for application and its
reconstruction.

ev’_app : eval’ (app’ M1 M2) V

<- eval’ M1 (lam’ M1’)

<- eval’ M2 V2

<- eval’ (M1’ V2) V.

ev’_app :

{T1:tp} {T2:tp} {M1’:term T2 -> term T1} {V2:term T2} {V:term T1}

{M2:term T2} {M1:term (=> T2 T1)}

eval’ T1 (M1’ V2) V -> eval’ T2 M2 V2

-> eval’ (=> T2 T1) M1 (lam’ T2 T1 M1’)

-> eval’ T1 (app’ T2 T1 M1 M2) V.

One should keep in mind that such an indexed representation is not always
possible. The let name-construct from the original formulation of Mini-ML is an
example. The way it encodes polymorphism relies on multiple types for the same
expression, which cannot be encoded directly in indexed form. Systems that in-
corporate subtyping provide another class of examples that may be difficult or
impossible to represent in an indexed manner.


